

# NAG Library Routine Document

## G05TLF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G05TLF generates a vector of pseudorandom integers uniformly distributed over the interval  $[a, b]$ .

### 2 Specification

SUBROUTINE G05TLF (N, A, B, STATE, X, IFAIL)

INTEGER N, A, B, STATE(\*), X(N), IFAIL

### 3 Description

G05TLF generates the next  $n$  values  $y_i$  from a uniform  $(0, 1]$  generator (see G05SAF for details) and applies the transformation

$$x_i = a + \lfloor (b - a + 1)y_i \rfloor,$$

where  $\lfloor z \rfloor$  is the integer part of the real value  $z$ . The routine ensures that the values  $x_i$  lie in the closed interval  $[a, b]$ .

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05TLF.

### 4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

### 5 Parameters

1: N – INTEGER *Input*

*On entry:*  $n$ , the number of pseudorandom numbers to be generated.

*Constraint:*  $N \geq 0$ .

2: A – INTEGER *Input*

3: B – INTEGER *Input*

*On entry:* the end points  $a$  and  $b$  of the uniform distribution.

*Constraint:*  $A \leq B$ .

4: STATE(\*) – INTEGER array *Communication Array*

**Note:** the actual argument supplied must be the array STATE supplied to the initialization routines G05KFF or G05KGF.

*On entry:* contains information on the selected base generator and its current state.

*On exit:* contains updated information on the state of the generator.

5: X(N) – INTEGER array *Output*

*On exit:* the  $n$  pseudorandom numbers from the specified uniform distribution.

## 6: IFAIL – INTEGER

*Input/Output*

*On entry:* IFAIL must be set to 0,  $-1$  or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value  $-1$  or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value  $-1$  or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $N < 0$ .

IFAIL = 3

On entry,  $B < A$ .

IFAIL = 4

On entry, STATE vector was not initialized or has been corrupted.

## 7 Accuracy

Not applicable.

## 8 Further Comments

None.

## 9 Example

This example prints five pseudorandom integers from a discrete uniform distribution between  $-5$  and 5, generated by a single call to G05TLF, after initialization by G05KFF.

### 9.1 Program Text

```

Program g05tlfe
!      G05TLF Example Program Text
!
!      Mark 24 Release. NAG Copyright 2012.
!
!      .. Use Statements ..
Use nag_library, Only: g05kff, g05tlf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: lseed = 1, nin = 5, nout = 6
!      .. Local Scalars ..
Integer                    :: a, b, genid, ifail, lstate, n, subid
!      .. Local Arrays ..
Integer                    :: seed(lseed)

```

```

Integer, Allocatable          :: state(:), x(:)
! .. Executable Statements ..
Write (nout,*) 'G05TLF Example Program Results'
Write (nout,*)

! Skip heading in data file
Read (nin,*)

! Read in the base generator information and seed
Read (nin,*) genid, subid, seed(1)

! Initial call to initialiser to get size of STATE array
lstate = 0
Allocate (state(lstate))
ifail = 0
Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

! Reallocate STATE
Deallocate (state)
Allocate (state(lstate))

! Initialize the generator to a repeatable sequence
ifail = 0
Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

! Read in sample size
Read (nin,*) n

! Read in the distribution parameters
Read (nin,*) a, b

Allocate (x(n))

! Generate the variates
ifail = 0
Call g05tlf(n,a,b,state,x,ifail)

! Display the variates
Write (nout,99999) x(1:n)

99999 Format (1X,I12)
End Program g05tlfe

```

## 9.2 Program Data

```

G05TLF Example Program Data
1 1 1762543          :: GENID,SUBID,SEED(1)
5                  :: N
-5 5               :: A,B

```

## 9.3 Program Results

```

G05TLF Example Program Results

      2
     -4
      3
      3
     -4

```

---